// File: contracts/Wrapper.sol



//                                 ▄,
//                       ▄        █▌█▌
//                     ╓██      ,█▀█▌
//                    ▄██▌    ▄█▀' █▌          ███
//                    ▓█ ╙███▀▀└    ╙█µ       ╓█▀█─
//                  █▌               ▀█▄╓,,▄█▀ ╫▌
//                 █▌                   ╙╙╙'   █▌
//                ▐█                           ╟█    ▓█
//               ┌███▀▀▀▀▀▀▀██▄▄,               █▄ ╓██⌐
//               █▀            '╙▀▀█▄▄           ╙▀▀█▌
//              ╔▌                   ╙▀█▄,         ▐█
//              █─                       ╙██,     ┌█⌐
//             ▐█                           ▀█▄   █▀
//             ╟█                             ▀█ █▀
//            ╓█▀                              ╙█
//            █─    ▄█▀▀▀▀▀██▌   ▐█▄▄▄▄▄       ▄█
//            █     ╙▀▀▀▀   ╙╓▄▄ └▀└  ╓└▀█▄    ╙█
//            █▌            █▀└└▀█    ╙▀▀▀└    ▐█
//             █▌           ╙   j█µ            █▌
//             ╟▌       ,███▄▄,╓,▐█          ,█▀
//             █▌      ╒█─ ▄▄▄╠▄╙▀'          █▌
//             █⌐      ███▀╙,╙╙╙▀█▄          █
//             █▌      └┌█▀▀▀██▀  ██        ╟█
//             ╙█        ▀█▄▄▄▄▄▄█▀`       ╓█
//              ╙█▄         '└└└         ,█▀
//                ╙▀█▄   ▀▀▀▀▀▀██     ,▄█▀
//        __      __ ╙▀█▄,         ▄▓█▀'
//       / /      \ \    └╙▀▀▀██▀▀▀▀╙
//      | |_      _| |_ __ __ _ _ __  _ __   ___ _ __ 
//      | \ \ /\ / / | '__/ _` | '_ \| '_ \ / _ \ '__|
//      | |\ V  V /| | | | (_| | |_) | |_) |  __/ |   
//      | | \_/\_/ | |_|  \__,_| .__/| .__/ \___|_|   
//       \_\      /_/          | |   | | ғɪᴠᴇ ʟɪɴᴇs             
//                             |_|   |_|              
//
//               SPDX-License-Identifier: MIT
//              Written by Buzzy @ buzzybee.eth
pragma solidity ^0.8.7;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC721/IERC721Receiver.sol";
import "@openzeppelin/contracts/token/ERC1155/utils/ERC1155Holder.sol";

error Rejected721();

contract Wrapper is Ownable, IERC721Receiver, ERC1155Holder {
    address public openstore;
    address public fivelines;
    address private _devon = 0xfCdB35c1105cA8EA01Df3B81A4570FF621817Cb8;
    address private _buzz = 0x816ae721F90d9cd5190d0385E7224C6798DaD52B;

    mapping(uint256 => bool) public wrapped;
    mapping(uint256 => uint256) public tokenMapping;

    uint256[] private tokens;

    constructor(
        address _openstore,
        address _fivelines,
        uint256[] memory openseaTokenIds,
        uint256[] memory fivelinesTokenIds
    ) {
        openstore = _openstore;
        fivelines = _fivelines;

        for (uint256 i = 0; i < openseaTokenIds.length; i++) {
            tokenMapping[openseaTokenIds[i]] = fivelinesTokenIds[i];
        }
    }

    function setTokenMappings(
        uint256[] calldata openseaTokenIds,
        uint256[] calldata fivelinesTokenIds
    ) public onlyOwner {
        for (uint256 i = 0; i < openseaTokenIds.length; i++) {
            tokenMapping[openseaTokenIds[i]] = fivelinesTokenIds[i];
        }
    }

    function setFiveLinesContractAddress(address _fivelines) public onlyOwner {
        fivelines = _fivelines;
    }

    function setOpenstoreContractAddress(address _openstore) public onlyOwner {
        openstore = _openstore;
    }

    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) public view override returns (bytes4) {
        // this doesn't work alongside manifold's batch minter.
        // if we had more time, I'd make it all work perfectly,
        // but the bottom line is I can't quite keep track of their
        // batch mint address as well as I'd like. For the time being,
        // lets just accept all 721s!
        // require(operator == fivelines || from == _buzz, "Token is not from the Five Lines collection");
        require(!wrapped[tokenId], "Token is already wrapped");
        return this.onERC721Received.selector;
    }

    function _transferWrapped(address to, uint id) private {
        require(!wrapped[id], "Token is already wrapped");

        (bool success,) = fivelines.call(
            abi.encodeWithSignature(
                "safeTransferFrom(address,address,uint256)",
                address(this),
                to,
                id
            )
        );

        require(success, "Transfer failed");

        wrapped[id] = true;
    }

    function _handleOpenstoreReceive(address operator, uint256 id) private {
        require(tokenMapping[id] != 0, "Token ID not in mapping, is this a valid Five Lines token?");
        tokens.push(id);

        _transferWrapped(operator, tokenMapping[id]);
    }

    function onERC1155Received(
        address operator,
        address from,
        uint256 id,
        uint256 value,
        bytes memory data
    ) public override returns (bytes4) {
        require(msg.sender == openstore, "ERC1155 not from Opensea's shared contract");
        _handleOpenstoreReceive(operator, id);
        return this.onERC1155Received.selector;
    }

    function onERC1155BatchReceived(
        address operator,
        address from,
        uint256[] memory ids,
        uint256[] memory values,
        bytes memory data
    ) public override returns(bytes4) {
        require(msg.sender == openstore, "ERC1155 batch not from Opensea's shared contract");
        for (uint256 i; i < ids.length; i++) {
            _handleOpenstoreReceive(operator, ids[i]);
        }
        return this.onERC1155BatchReceived.selector;
    }

    function withdraw() public onlyOwner {
        while (tokens.length > 0) {
            uint256 token = tokens[tokens.length - 1];
            (bool success,) = openstore.call(
                abi.encodeWithSignature(
                    "safeTransferFrom(address,address,uint256,uint256,bytes)",
                    address(this),
                    _devon,
                    token,
                    1,
                    ""
                )
            );

            require(success, "Token transfer failed");

            tokens.pop();
        }
    }
}
// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC1155/utils/ERC1155Holder.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC1155/utils/ERC1155Holder.sol)

pragma solidity ^0.8.0;

import "./ERC1155Receiver.sol";

/**
 * Simple implementation of `ERC1155Receiver` that will allow a contract to hold ERC1155 tokens.
 *
 * IMPORTANT: When inheriting this contract, you must include a way to use the received tokens, otherwise they will be
 * stuck.
 *
 * @dev _Available since v3.1._
 */
contract ERC1155Holder is ERC1155Receiver {
    function onERC1155Received(
        address,
        address,
        uint256,
        uint256,
        bytes memory
    ) public virtual override returns (bytes4) {
        return this.onERC1155Received.selector;
    }

    function onERC1155BatchReceived(
        address,
        address,
        uint256[] memory,
        uint256[] memory,
        bytes memory
    ) public virtual override returns (bytes4) {
        return this.onERC1155BatchReceived.selector;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC1155/utils/ERC1155Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC1155/utils/ERC1155Receiver.sol)

pragma solidity ^0.8.0;

import "../IERC1155Receiver.sol";
import "../../../utils/introspection/ERC165.sol";

/**
 * @dev _Available since v3.1._
 */
abstract contract ERC1155Receiver is ERC165, IERC1155Receiver {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return interfaceId == type(IERC1155Receiver).interfaceId || super.supportsInterface(interfaceId);
    }
}

// File: @openzeppelin/contracts/token/ERC1155/IERC1155Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC1155/IERC1155Receiver.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev _Available since v3.1._
 */
interface IERC1155Receiver is IERC165 {
    /**
     * @dev Handles the receipt of a single ERC1155 token type. This function is
     * called at the end of a `safeTransferFrom` after the balance has been updated.
     *
     * NOTE: To accept the transfer, this must return
     * `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))`
     * (i.e. 0xf23a6e61, or its own function selector).
     *
     * @param operator The address which initiated the transfer (i.e. msg.sender)
     * @param from The address which previously owned the token
     * @param id The ID of the token being transferred
     * @param value The amount of tokens being transferred
     * @param data Additional data with no specified format
     * @return `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))` if transfer is allowed
     */
    function onERC1155Received(
        address operator,
        address from,
        uint256 id,
        uint256 value,
        bytes calldata data
    ) external returns (bytes4);

    /**
     * @dev Handles the receipt of a multiple ERC1155 token types. This function
     * is called at the end of a `safeBatchTransferFrom` after the balances have
     * been updated.
     *
     * NOTE: To accept the transfer(s), this must return
     * `bytes4(keccak256("onERC1155BatchReceived(address,address,uint256[],uint256[],bytes)"))`
     * (i.e. 0xbc197c81, or its own function selector).
     *
     * @param operator The address which initiated the batch transfer (i.e. msg.sender)
     * @param from The address which previously owned the token
     * @param ids An array containing ids of each token being transferred (order and length must match values array)
     * @param values An array containing amounts of each token being transferred (order and length must match ids array)
     * @param data Additional data with no specified format
     * @return `bytes4(keccak256("onERC1155BatchReceived(address,address,uint256[],uint256[],bytes)"))` if transfer is allowed
     */
    function onERC1155BatchReceived(
        address operator,
        address from,
        uint256[] calldata ids,
        uint256[] calldata values,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}